# Maintainer: Felix Yan <felixonmars@archlinux.org>
# Contributor: Vojtěch Aschenbrenner <v@asch.cz>

pkgname=minisat
pkgver=2.2.1.20201216
_commit=55d7e7613b94f0369476a0a8ff35969485ed324f
pkgrel=1
pkgdesc="A minimalistic and high-performance SAT solver"
arch=('loong64' 'x86_64')
url="http://minisat.se/"
license=('MIT')
depends=('gcc-libs' 'zlib')
source=(https://github.com/msoos/minisat/archive/$_commit/$pkgname-$_commit.tar.gz)
sha512sums=('17ec01c5083d91f55b8ccc14176b673db225156c0270762fdf154552f95ece78d31d7017a3ce5d54831859a3f5af45ac9f465bd590a24cb0f7dcc3dfb33fa8d9')

build() {
  cd $pkgname-$_commit
  make config prefix=/usr
  make
}

package() {
  cd $pkgname-$_commit
  make install DESTDIR="$pkgdir"
  install -Dm644 LICENSE -t "$pkgdir"/usr/share/licenses/$pkgname/
}

# vim:set ts=2 sw=2 et:
